
Supplemental Info 1. JAGS Code for the Pollock’s Robust Design model used in this study.  

# Model specification ---- 
modelString = " 
model { 
  # Priors 
    # Survival and emigration 
      for(i in 1:(n.ind)){ 
        for(t in 1:(n.years-1)){ 
          # Logit-scale predictor of apparent survival (species) 
            logit(r_phi[i,t]) <- l_phi_i[i,t] 
            l_phi_i[i,t] <-  l_phi[species[i],t] 
           
          # Logit-scale predictor of site fidelity (species) 
            logit(r_gamma[i,t]) <- l_gamma_i[i,t] 
            l_gamma_i[i,t] <- l_gamma[species[i],t] 
        } 
      } 
     
    # Species- and time-specific priors on 
    # phi and gamma 
      for(s in 1:n.species){ 
        for(t in 1:(n.years-1)){ 
          l_phi[s,t] ~ dnorm(0, 1) 
          l_gamma[s,t] ~ dnorm(0, 1) 
 
          logit(phi[s,t]) <- l_phi[s,t] 
          logit(gamma[s,t]) <- l_gamma[s,t] 
        } 
      } 
     
    # Secondary occasions p 
      # Logit-scale linear predictor on 
      # p by year and secondary occasion 
        for (i in 1:n.ind){ 
          for (t in 1:n.years){ 
            for (j in 1:(n.sec[t])){ 
                logit(p[i,t,j]) <- lp[i,t,j]   
                lp[i,t,j] <- inprod(beta[t,j, ], X[i, ]) 
                yes[i,t,j] ~ dbin(p[i,t,j], total[i,t,j]) 
            } 
          } 
        } 
      # Betas (intercepts & 'slopes') for linear predictor of logit-scale 
      # p, varying by time-period 
      for(t in 1:n.years){ 
        for(j in 1:(n.sec[t])){ 



          for(n in 1:nX){ 
            beta[t,j,n] ~ dnorm(0, 0.001) 
          } 
        } 
      } 
 
    # Primary occasions p 
      for(i in 1:n.ind){ 
        for(t in 1:n.years){         
          pstar[i,t] <- (1 - prod(1 - p[i,t,]) ) * (1 - p_loss[i]) 
        } 
      } 
   
  # State (presence/absence) and observation (detection) 
  # process likelihoods 
    for (i in 1:n.ind){ 
 
      # Initialize first capture 
        z[i,first[i]] <- ch[i,first[i]] 
   
      # Likelihood for state process 
        for (t in (first[i]+1):n.years){ 
          mu1[i,t] <- z[i,t-1] * r_phi[i, t-1] 
          mu2[i,t] <- z[i,t] * r_gamma[i, t-1] * pstar[i,t] 
          z[i,t] ~ dbern(mu1[i,t]) 
          ch[i,t] ~ dbern(mu2[i,t]) 
        } 
    } 
     
  # Count process likelihood (abundance) 
  # Population abundance estimated using n-mixture 
  # approach. We assume that observed abundance 
  # is the outcome (C) of a binomial process with some 
  # unknown probability of success (pstar) and some 
  # starting number of trials (N) 
  for(s in 1:n.species){ 
    # Initialize counts for each species  
    N[s, 1] ~ dpois(maxes[s]) 
 
    log(lambda[s]) <- mu[s] 
    mu[s] ~ dunif(-10, 10) 
 
    # Total number alive at time t as function of 
    # number alive at t-1, survival, and site fidelity 
    for(t in 2:n.years){ 
      N[s,t] ~ dpois(N[s, t-1]*phi[s, t-1]*gamma[s,t-1]) 



    } 
 
    for(t in 1:n.years){ 
      C[s, t] ~ dpois(N[s,t]/mean(pstar[,t])) 
    } 
 
  } 
   
} 
" 
  



Supplemental Table 1. Secondary period detection probabilities for the upper and lower sites in 
the Colorado River. Presented are secondary period detection probability estimates, along with 
95% credible intervals (lower bound (LB), upper bound (UB)).    

Site Primary Period Secondary Period Estimate LB UB 
Upper 1 1 0.502 0.496 0.513 

 1 2 0.502 0.496 0.513 

 1 3 0.502 0.496 0.513 

 2 1 0.502 0.492 0.507 

 2 2 0.502 0.492 0.507 

 2 3 0.502 0.492 0.507 

 3 1 0.502 0.491 0.512 

 3 2 0.502 0.491 0.512 

 3 3 0.502 0.491 0.512 

 4 1 0.502 0.497 0.507 

 4 2 0.502 0.497 0.507 

 4 3 0.502 0.497 0.507 

 5 1 0.502 0.489 0.509 

 5 2 0.502 0.489 0.509 

 5 3 0.502 0.489 0.509 
Lower 1 1 0.365 0.345 0.428 

 1 2 0.365 0.354 0.403 

 1 3 0.365 0.355 0.408 

 2 1 0.363 0.333 0.383 

 2 2 0.363 0.334 0.381 

 2 3 0.363 0.330 0.381 

 3 1 0.364 0.359 0.367 

 3 2 0.364 0.358 0.369 

 3 3 0.364 0.345 0.368 

 4 1 0.364 0.343 0.401 

 4 2 0.364 0.340 0.403 

 4 3 0.364 0.347 0.398 

 5 1 0.363 0.314 0.385 

 5 2 0.363 0.319 0.373 

 5 3 0.363 0.315 0.382 
 

  



Supplemental Table 2. Tagging configuration detection probabilities for the upper and lower 
sites in the Colorado River. Presented are primary period detection probability estimates, along 
with 95% credible intervals (lower bound (LB), upper bound (UB)).    

Site Tagging Configuration Estimate LB UB 
Upper 1 shellfish 0.855 0.847 0.862 

 1 shellfish/1 PIT  0.875 0.872 0.879 

 2 shellfish 0.876 0.868 0.883 

 2 shellfish/1 PIT 0.876 0.873 0.879 
Lower 1 shellfish 0.723 0.665 0.778 

 1 shellfish/1 PIT  0.742 0.736 0.748 

 2 shellfish 0.742 0.720 0.762 

 2 shellfish/1 PIT 0.730 0.565 0.857 
 


